

HB-25 Motor Controller (#29144)

General Description

The HB-25 motor controller combines the power of an H-bridge with the simplicity of a servo. The HB-25 is more than just a motor driver chip connected to some logic and high current on a PCB. A quality H-bridge involves an efficient thermal design, which is what the HB-25 has accomplished through a machined heat sink and fan to draw air over the H-bridge. Additionally, it uses a thermal bonding agent to transfer heat from the motor driver chip to the heat sink. The result is a high-current motor controller with great thermal characteristics and requiring no additional hardware for cooling.

Before using your HB-25 Motor Controller, read and understand this entire document, including the Precautions section beginning on page 4.

Product Update Notice: See Initialization Window on page 4 for details.

Features

- 25 A continuous current, 35 A surge @ 13.8 VDC
- Works with any size motor up to ½ HP
- Control a DC motor just like a continuous rotation servo
- A single pulse required to set motor speed
- Compatible with all Parallax microcontrollers
- 2 operation modes: control 1 or 2 HB-25's independently from a single I/O line
- Built-in automatic shut-off if invalid pulse widths are received
- Communication Timeout mode option for automatic shutoff

Application Ideas

- Robotics
- Automotive Applications


Resources and Downloads

Check out the HB-25 Motor Controller product page for the example source code and other resources:

<http://www.parallax.com/product/29144>

Interface Connections

Vin	Power input terminal, +6 to +16 VDC
GND	Negative side of battery terminal
M1 & M2	Motor connectors (Polarity reverses direction)
W	Servo pulse input
R	Not connected
B	Servo ground
J	Mode jumper pins



Modes of Operation

The HB-25 connects to the microcontroller much like a servo. You can use an extension cable (such as Parallax part #805-00012) or a custom cable to connect the HB-25 to your controller. Reversing the M1 and M2 connections to the motor effectively reverses the direction. The HB-25 has two modes of operation which are selected by the jumper labeled "J".

Mode 1 is Single Mode, where only one HB-25 is present on the microcontroller I/O line. In this mode, the HB-25 can be controlled with as little as a single pulse from your microcontroller.

Mode 2 is dual-mode, which is used when connecting a second HB-25 to a first HB-25, rather than connecting directly to the microcontroller. This powerful feature allows two HB-25's to be independently controlled through the same I/O line, saving I/O pins. Two HB-25's are required to use Mode 2.

Mode 1

Mode 1 is selected when the jumper labeled "J" is in place.




Figure 1: Single Mode Selection

Mode 1 Communication

In this mode, a single pulse value sent to the HB-25 can control the HB-25; no "refreshing" is necessary, as is the case with a servo. For compatibility however, you may send the HB-25 pulses every 20 ms just like a servo and it will function the same.

There is a hold-off time of 5 ms where the HB-25 will ignore incoming pulses. As a result, the unit should not be refreshed more frequently than about 5.25 ms + pulse time. Pulse time can be anywhere from 0.8 ms to 2.2 ms. If the HB-25 receives a pulse outside of this range, it will temporarily shut off the motor until it receives a valid pulse.




Figure 2: Single Mode Communication

The maximum time between pulses is unlimited, since a single pulse is all that is required to set the unit running indefinitely. This is especially nice for BASIC Stamp users with complex sensor code which may not have had sufficient time to refresh servos during the main loop. You only need to send a new pulse when you are ready to change the speed of the motor (or stop it).

A situation in which refresh pulses are necessary occurs when the Communication Timeout feature is enabled, as described on page 5.

Mode 2

Mode 2 is selected when the jumper is removed. Referring to Figure 3, Mode 2 should only be selected on Unit 2, which is the unit connected only to the other HB-25. Unit 1 needs to remain in Mode 1, with its jumper left in place.




Figure 3: Dual Mode

The jumper is removed from Motor 2, and is in place in Motor 1

Mode 2 Communication

Independent control of both HB-25 units through a single I/O line is accomplished by sending two sequential pulses with a pause of 1.1 ms between them. When in Mode 2, the HB-25 looks for two sequential pulses and only responds to the second one. In reference to the diagram in Figure 4, Unit 1 would respond to Pulse #1, and Unit 2 would respond to Pulse #2. Unit 2 looks for Pulse #1, pauses for a 1 ms hold-off time, then looks for Pulse #2. Therefore, the timing between Pulse #1 and Pulse #2 is important, and there should be a minimum of 1.1 ms between pulses.




Figure 4: Dual Mode Communication

Just as in Mode 1, the HB-25 observes a 5 ms hold-ff time after the second pulse, so there should be a 5.25 ms pause after the second pulse. If a pulse were to start before the end of this hold-off time it might be seen as a shorter pulse than it is since the first part was cut off. This could cause a 2 ms pulse intended to be full speed in one direction to be interpreted as a shorter pulse, such as 1 ms, causing the motor to turn full speed in the opposite direction! Obviously this is not desirable, so observing the hold-off time is important to prevent this.

As a result, the unit should not be refreshed more frequently than about 5.25 ms + 1.1 ms + Pulse #1 + Pulse #2. Pulse time can be anywhere from 0.8 ms to 2.2 ms. If the HB-25 receives a pulse outside of this range, it will temporarily shut off the motor until it receives a valid pulse.

As with Mode 1, since each motor only requires a single pulse to set it indefinitely, there is no refreshing required unless the Communication Timeout feature is enabled.

BASIC Stamp 1 and 2 programmers using our example code will find that a PAUSE 1 command between Pulse #1 and Pulse #2 and a PAUSE 5 after Pulse #2 are adequate. This is because code execution time extends the resulting low times just enough to exceed the HB-25's own hold-off times. Those using faster BS2 models or microcontrollers such as the SX will need to observe the minimum pause times more strictly.

Initialization Window

Regardless of the mode the HB-25 signal pin should be brought low immediately upon power-up. This initialization window is 2 seconds for Rev F firmware released March 2014 (dot on 8-pin IC) and 50 ms for previous versions of firmware (no dot on 8-pin IC). This means for versions *prior* to Rev F you may have to power-up the HB-25 after your MCU to ensure it can initialize the signal pin in time. Optionally you could install a 4.7 kΩ resistor between ground and the signal pin to ensure the line is brought low on power-up. This resistor would need to be removed prior to changing the Timeout Mode.

Specifications

Motor Size	Any motor up to ½ HP max
Motor Supply	6.0 VDC – 16.0 VDC
Logic Supply	N/A – Internal Regulator
Load Current	25A Continuous 35A Surge (13.8 V)
Standby Current	50 mA @ 6 V, 80 mA @ 13.8 V (fan on)
PWM Frequency	9.2 kHz
Pulse Input	1.0 ms Full Reverse 1.5 ms Neutral (Off) 2.0 ms Full Forward
Pulse Refresh Rate	Not Required, Single Pulse Operation Capable
Modes	Single/Dual Motor Control
Protection Circuits	Over Voltage, Over Current, Over Temp
Automatic Fault Reset Indicators	Power (Green), Fault (Red)
Fuse	Mini ATC Standard 25 amp maximum
Cooling	Forced Air – Ball Bearing Fan
Terminals	Screw Posts with 35 A Rating
Weight	2.5 oz (71 grams)
Size	1.6" x 1.6" x 1.9"
Mounting	2 ea. 6-32 screws on .800" centers

Fuses

The HB-25 uses Mini ATC Standard fuses available from automotive supply sources. Always use the appropriate size fuse for the motor you are using. This can be determined by finding out the stall current of your motor. For example, your motor may draw only 5 to 10 amps during normal use but could draw 15 amps when stalled. In this case you would replace the 25 amp fuse with the 15 amp fuse. What this will do is protect your motor in case of a stall by blowing the fuse instead of the motor burning itself up.

Precautions

Testing for Reverse Polarity

There is *no reverse polarity protection* on the power input! If you are unsure of your wiring, before applying power do the following:

1. REMOVE THE FUSE from the HB-25. ***This is very important!***
2. BRIEFLY apply power to the unit.
3. If the Green Power LED did not illuminate, DISCONNECT POWER IMMEDIATELY as this indicates incorrect polarity; troubleshoot your wiring before continuing.
4. If the Green Power LED illuminated the polarity is correct.
5. Immediately remove power after verifying polarity. If the polarity is incorrect leaving power connected could overheat the power-sensing components.
6. REPLACE THE FUSE before continuing.

Safe Power-Up and Power-Down

Remember, because the HB-25 responds to pulses like a servo, changing states on your microcontroller's I/O pins at startup and shutdown can affect the HB-25 and set it into an unintended state. This includes the I/O pins changing between input and output as well as between output high and output low.

If your microcontroller is powered up or powered down while the HB-25 is powered up, the HB-25 could receive a false trigger. To prevent this, use one of the following options:

- Most reliable option:
 1. Power the microcontroller up first, using code that forces the program to wait until the HB-25 is powered up before sending pulses. Code examples are provided below.
 2. Power up the HB-25 for operation.
 3. When shutting down your application, cut off power to the HB-25 first.
 4. Power down your microcontroller last.
- Or, power the HB-25 and the microcontroller at the same time using a double-post (dual-circuit) switch.
- If there is the possibility that the HB-25 will be on before your microcontroller, use the example code as mentioned above to ensure that the HB-25 is properly initialized when the microcontroller is powered on. The I/O line connected to the HB-25 should be made LOW either before or immediately after the HB-25 powers up. If this does not happen the HB-25 may fail to respond to incoming pulses.

If your application poses any potential safety risk and you need the HB-25 to shut off the motors when the microcontroller has lost power or has malfunctioned, enable the Timeout Feature described below before operation.

Communication Timeout Mode

The HB-25 has a selectable Timeout mode. When this mode is enabled, the HB-25 will shut off the motor after 4 seconds if it does not receive pulses from the microcontroller. Once the motor is shut off, if the HB-25 receives a valid pulse again, it will restart the motor. Timeout mode should always be enabled in applications where a failure of your microcontroller could cause a dangerous situation or safety risk, or potentially cause damage to your application.

When this mode is enabled, the HB-25 will require refreshing by the microcontroller like a servo. These pulses could be 20 ms – 50 ms apart as would be typical with a servo, but may be up to 4 seconds apart.


The Timeout mode can be toggled on and off by following the procedure below. The current state is stored in an on-board EEPROM and will remain until you toggle it again.

To toggle the state of the Timeout mode:

1. Disconnect the servo input cable from the HB-25. If there is a daisy-chained cable to a second unit remove it as well. It makes no difference whether the Mode Jumper (J) is installed.
2. Apply power to the HB-25 and wait 4 seconds.
3. With the power on change the state of the Mode Jumper by installing the jumper if it is removed or removing it if it is installed.
4. Remove and re-apply power.

The timeout state should now be toggled. To change it back, repeat the above steps.

Module Dimensions



Example Source Code

BASIC Stamp® 1 Program

Before using this example code, read this entire document, including the Precautions section. This code is for testing 1 HB-25 connected to a BASIC Stamp 1 I/O Pin P0.

```
' =====
' File..... HB-25 Motor Test.bsl
' Purpose... Tests One HB-25 Connected To P0
' Author.... Parallax, Inc.
' E-mail.... support@parallax.com
' Updated... 01-18-2006

' {$STAMP BS1}
' {$PBASIC 1.0}

' -----[ Program Description ]-----

' This program tests the HB-25 by waiting for it to power up, then pulsing
' the output to ramp the motor up, wait 3 seconds then ramp back down to a
' stop. This code was written for one HB-25 connected in Mode 1 (Single)
' to PIN0.

' -----[ I/O Definitions ]-----

SYMBOL          HB25 =  PIN0          ' I/O Pin For HB-25

' -----[ Variables ]-----

SYMBOL          index = B0            ' Variable Space For Ramp Counter

' -----[ Initialization ]-----

Startup:
  IF HB25 = 0 THEN Startup             ' Wait For HB-25 Power Up
  LET          DIRS = %00000001       ' Set PIN0 To Output
  PAUSE 5                               ' Wait For HB-25 To Initialize
  PULSOUT HB25, 150                    ' Stop Motor
  PAUSE 20                             ' Wait 20 mS

' -----[ Program Code ]-----

Main:
  FOR index = 150 TO 200               ' Ramp Up To Full Speed
    PULSOUT HB25, index               ' Motor Forward
    PAUSE 150                         ' 150 mS Smoothing Delay
  NEXT
  PAUSE 3000                          ' Wait 3 Seconds
  FOR index = 200 TO 150 STEP -1      ' Ramp Back Down
    PULSOUT HB25, index               ' Motor Forward Slowing
    PAUSE 150                         ' 150 mS Smoothing Delay
  NEXT

Finish:
  GOTO Finish                         ' Prevent PIN0 From Changing State
```

BASIC Stamp 2 Program

Before using this example code, read this entire document, including the Precautions section. This code is for testing 1 or 2 HB-25 Motor Controllers connected to a BASIC Stamp 2 I/O pin P15.

```
' =====
' File..... HB-25 Motor Test.bs2
' Purpose... Tests One Or Two HB-25's Connected To P15
' Author.... Parallax, Inc.
' E-mail.... support@parallax.com
' Updated... 01-18-2006
'
' {$STAMP BS2}
' {$PBASIC 2.5}

' -----[ Program Description ]-----

' This program tests the HB-25 by waiting for it to power up, then pulsing
' the output to ramp the motors up in opposite directions, wait 3 seconds
' then ramp them back down to a stopped position. While the code is
' written for two HB-25/motors you can use it with just one by commenting
' out or removing the lines for the second motor, commented below. If you
' have two HB-25/motors connected, remember to remove the jumper block from
' the second HB-25.

' -----[ I/O Definitions ]-----

HB25          PIN      15          ' I/O Pin For HB-25

' -----[ Variables ]-----

index          VAR      Word        ' Counter For Ramping

' -----[ Initialization ]-----

DO : LOOP UNTIL HB25 = 1          ' Wait For HB-25 Power Up
LOW HB25                          ' Make I/O Pin Output/Low
PAUSE 5                          ' Wait For HB-25 To Initialize
PULSOUT HB25, 750                 ' Stop Motor 1
PAUSE 1                          ' 1 mS Delay
PULSOUT HB25, 750                 ' Stop Motor 2 (If Connected)
                                ' The Above 2 Lines May Be Removed
                                ' If You Are Using Single Mode

' -----[ Program Code ]-----

Main:
  PAUSE 20                        ' Wait 20 mS Before Ramping
  FOR index = 0 TO 250            ' Ramp Up To Full Speed
    PULSOUT HB25, 750 + index     ' Motor 1 Forward
    PAUSE 1                      ' 1 mS Delay For Motor 2 Pulse
    PULSOUT HB25, 750 - index     ' Motor 2 Reverse
    PAUSE 20                     ' 20 mS Smoothing Delay
  NEXT
  PAUSE 3000                     ' Wait 3 Seconds
  FOR index = 250 TO 0           ' Ramp Back Down
    PULSOUT HB25, 750 + index     ' Motor 1 Forward Slowing
    PAUSE 1                      ' 1 mS Delay For Motor 2
    PULSOUT HB25, 750 - index     ' Motor 2 Reverse Slowing
    PAUSE 20                     ' 20 mS Smoothing Delay
  NEXT
  STOP                          ' Use STOP To Prevent State Change
```

Revision History

Version 1.3: Product Update Notice added to first page, explained in Initialization Window section added on page 4.